

Programa SAS para Análise de Tabelas de Vida e Fertilidade de Artrópodes: o Método Jackknife

Aline de Holanda Nunes Maia¹
Alfredo José Barreto Luiz²

Introdução

O conhecimento sobre o potencial de crescimento de populações de artrópodes é fundamental para o estabelecimento de práticas eficientes de controle de pragas e para a avaliação de possíveis impactos ambientais de tecnologias sobre artrópodes não-alvo (Bleicher & Parra, 1990; Fernandez-Casaldelrey, 1992; Brodsgaard, 1994; Sharma et al., 1994; Parra et al., 1995; Sharma et al., 1997; Nascimento et al., 1998). Um dos métodos tradicionalmente utilizados para estimação de taxas de crescimento populacional em artrópodes utiliza tabelas de vida e fertilidade que sintetizam dados de sobrevivência e fertilidade de populações. Os principais parâmetros associados às tabelas de vida e fertilidade (TBVF) são: taxa líquida de reprodução (R_0); taxa intrínseca de crescimento (R_m); intervalo médio entre gerações (IMG); tempo de duplicação (TD); e razão finita de crescimento (λ) (Southwood, 1978).

Métodos Monte Carlo, tais como Jackknife, Bootstrap e testes permutacionais, são técnicas utilizadas, entre outras finalidades, para comparação de grupos (tratamentos) com relação a taxas de crescimento populacional, quando estas são estimadas utilizando tabelas de vida. Esses métodos computacionalmente intensivos são adequados para estimar

variâncias de funções complexas de dados observados (Manly, 1991). Meyer et al. (1986) propuseram o uso dos métodos Jackknife e Bootstrap para quantificar a variância do estimador do R_m . O método Jackknife, proposto por Tukey (1958), é freqüentemente utilizado para quantificar a incerteza associada às estimativas de parâmetros, quando é difícil fazê-lo por meio de métodos analíticos. A base teórica do método jackknife é apresentada em Efron (1982).

Neste trabalho, serão discutidos aspectos teóricos relacionados à inferência sobre parâmetros associados a TBVF e aspectos operacionais do programa *tabela de vida.sas* que é uma versão em português do programa *lifetable.sas* (Maia et al., 2000), desenvolvido em ambiente SAS³, que utiliza o método Jackknife para comparação de parâmetros entre pares de grupos. É apresentado um exemplo de aplicação no qual os dados de sobrevivência e fertilidade de fêmeas classificadas em três grupos hipotéticos (1, 2 e 3). Os dados do exemplo foram gerados por simulação. O programa (**Anexo 1**, TabelaDeVida.sas) está disponível para download em <http://www.cnpma.embrapa.br/public/comunicado33.html> no link "Anexos". Os dados utilizados neste trabalho são apresentados no **Anexo 2** (dados Exemplo.xls).

¹ Engenheira Agrônoma, Doutora em Agronomia, Embrapa Meio Ambiente, Rod. SP 340, km 127,5 - Caixa Postal 69, Tanquinho Velho, 13.820-000 Jaguariúna, SP. ahmaia@cnpma.embrapa.br

² Engenheiro Agrônomo, Doutor em Sensoriamento Remoto, Embrapa Meio Ambiente, Rod. SP 340, km 127,5 - Caixa Postal 69, Tanquinho Velho, 13.820-000 Jaguariúna, SP. alfredo@cnpma.embrapa.br

³ O SAS é um sistema de "softwares" modular e integrado para gerenciamento e análise de informações. Inclui módulos de análise estatística, pesquisa operacional, controle de qualidade, econometria, geração de gráficos e relatórios, entre outros.

Dados para construção de uma tabela de vida e fertilidade

Os dados para estimar o potencial de crescimento populacional utilizando os métodos implementados no programa *tabela de vida.sas* devem ser provenientes de experimentos com delineamento inteiramente casualizado. Neste caso, *n* fêmeas adultas são alocadas aleatoriamente a *G* grupos. As fêmeas são acasaladas e observadas a intervalos de tempo constantes, geralmente dias ou semanas. É registrado o número de ovos postos por fêmea (*N_OVOS*), em cada data de oviposição, até sua morte. Os dados devem estar organizados em um arquivo do tipo planilha eletrônica (*.xls), que contenha as colunas descritas na Tabela 1.

Tabela 1. Descrição do formato do arquivo de dados para estimação de parâmetros associados às TBVF.

Coluna	Descrição
1	Identificação do grupo ou tratamento (<i>G</i>): variável numérica ou alfanumérica;
2	Identificação da fêmea (<i>f</i>): variável numérica;
3	Idade da fêmea em cada data de postura (<i>t</i>): variável numérica;
4	Número de ovos postos por fêmea em cada data de postura (<i>N_OVOS</i>): variável numérica;
5	Índice de sobrevivência na fase imatura (<i>SFimat</i>): variável numérica;
6	Proporção de fêmeas na população, usualmente denominada razão sexual (<i>PCT_FEM</i>): variável numérica.

Observação: dentro de cada coluna não deve haver espaço entre os caracteres.

O registro do número de ovos (*N_OVOS*) deve ser computado desde o início da idade adulta da fêmea mais precoce no grupo *G* até o dia da morte de cada fêmea. Mesmo as datas nas quais não houve oviposição devem ser incluídas no arquivo de dados, registrando-se o número de ovos como zero, para evitar erros no cálculo do número de fêmeas vivas em cada data. As fêmeas que não ovipositaram durante toda a vida devem também ser incluídas no conjunto de dados. A sua exclusão levaria à superestimação das taxa de crescimento populacional nos respectivos grupos. Para mais detalhes, ver o Anexo 2, (Exemplo.xls).

Detalhes do programa SAS

Identificação dos “produtos” SAS utilizados e fontes de informações

No cabeçalho do programa, são descritos os produtos e procedimentos do SAS System utilizados no programa, e os nomes dos arquivos com o programa SAS e com os dados (Figura 1). Os produtos são conjuntos de procedimentos - SAS Procedures ou PROCs - do SAS System (SAS, 1990). A informação sobre os métodos inferenciais utilizados neste trabalho está descrita em Maia et al. (2000).

No Sistema SAS®, são utilizadas três “janelas”: uma para edição dos códigos do programa (Program Editor Window), outra para visualização dos resultados das análises (Output Window) e outra para verificação de erros ocorridos durante a execução dos comandos (Log Window). Quando são executados comandos para construção de gráficos (Proc GPLOT), estes são apresentados numa quarta “janela” (Graph Window). Os gráficos gerados podem ser

exportados para arquivos gráficos do tipo BMP ou JPG, por exemplo.

```

/******
/*PROGRAMA SAS PARA ESTIMAR PARÂMETROS ASSOCIADOS A
/*TABELAS DE VIDA E FERTILIDADE
/*
/*NOME: tabela de vida.sas
/*DADOS: exemplo.xls
/*
/*
/*PRODUTOS: SAS Language, BASE SAS Procedures, SAS Macro Language
/*e SASGRAPH
/*
/*
/*PROCS: SORT, MEANS, GPLOT, PRINT
/*
/*
/*REF.: MAIA, A. DE H. N.; LUIZ, A. J. B.; CAMPANHOLA, C. Statistical
/*inference on associated fertility life table parameters using Jackknife
/*technique: computational aspects. Journal of Economic Entomology. v. 93,
/*n. 2, p. 511-518, 2000.
/******

```

Fig. 1. Cabeçalho do programa que estima os parâmetros associados a tabelas de vida e fertilidade, com identificação dos produtos SAS e fontes de dados utilizados.

Criação de biblioteca SAS e leitura do arquivo de dados

Nesta seção, são descritos os procedimentos para a criação de uma biblioteca (opcional) para armazenamento de arquivos secundários em forma de planilhas SAS que podem ser lidas posteriormente com o uso do comando DATA (Fig. 2). Devem ser informados o nome e o endereço da biblioteca. O nome deve ter no máximo oito dígitos e não devem ser utilizados acentos, cedilha nem caracteres especiais, como por exemplo: %, & ou \$. O endereço é o diretório onde os arquivos secundários serão alocados. O comando para criação de uma biblioteca é: **libname** tabvida “C:\nome do diretório”.

Para a importação do arquivo externo que contém os dados, usa-se o **PROC IMPORT** (Fig. 2). Diferentes tipos de arquivos podem ser utilizados tais como arquivo texto (*.txt), arquivos Excel (*.xls, ou *.csv), entre outros. Em OUT=DATA1 informa-se o nome do arquivo interno do SAS no qual os dados serão armazenados; em DATAFILE= “C:\nome do diretório\Exemplo.xls” o nome e a localização do arquivo que contém as informações a serem utilizadas na análise. A opção DBMS= é utilizada para informar o tipo de arquivo a ser lido. As opções disponíveis se encontram listadas na Tabela 2.

```

/******
/*Leitura do arquivo Excel2000 (*.xls) contendo os dados de sobrevivência e
/*fertilidade com as seguintes variáveis (colunas):
/*
/*GRUPO - nome ou número do "tratamento"
/*
/*N_FEMEA - no. de identificação de cada fêmea
/*
/*IDADE - idade da fêmea em cada data de oviposição
/*
/*N_OVOS - número de ovos/fêmea em cada data de oviposição
/*
/*PCT_FEM - porcentagem de fêmeas na população de descendentes
/*
/*SFIMAT - sobrevivência dos descendentes durante a fase imatura
/******
PROC IMPORT OUT=DATA1
DATAFILE= "C:\nome do diretório\exemplo.xls" /*nome do arquivo*/
DBMS=EXCEL2000 REPLACE; /*tipo do arquivo*/
GETNAMES=YES; /*opção para ler o nome das variáveis na 1ª. linha */
RUN;
options ls=70 nodate pageno=1; /*largura e numeração da página*/
RUN;
PROC PRINT data=DATA1;
var GRUPO N_FEMEA IDADE N_OVOS PCT_FEM SFIMAT;
RUN;

```

Fig. 2. Lista de variáveis constantes do arquivo de dados a ser importado e comandos SAS para o procedimento de importação (PROC IMPORT) e de listagem na janela “output” (PROC PRINT).

Tabela 2. Tipos de arquivos que podem ser importados ao se utilizar o PROC IMPORT do SAS System e respectivas opções DBMF.

Opção DBMF =	Tipo de arquivo	Extensão
ACCESS	Base de dados do Microsoft Access	.MDB
DBF	Arquivo DBASE	.DBF
WK1	Planilha do Lótus versão 1	.WK1
WK3	Planilha do Lótus versão 3	.WK3
WK4	Planilha do Lótus versão 4	.WK4
EXCEL4	Planilha do Excel versão 4	.XLS
EXCEL5	Planilha do Excel versão 5	.XLS
EXCEL97	Planilha do Excel 97	.XLS
EXCEL2000	Planilha do Excel 2000	.XLS
DLM	Arquivo texto com colunas delimitadas por espaço	.
CSV	Arquivo texto com colunas delimitadas vírgulas	.CSV
TAB	Arquivo texto com colunas delimitadas por tabulações	.TAB

Padrões temporais de oviposição

A representação gráfica dos padrões de oviposição ao longo do tempo permite visualizar vários aspectos relacionados à reprodução das fêmeas em cada grupo: a) início, fim e duração do período de oviposição; b) duração dos períodos de pré e pós oviposição; c) duração da fase adulta das fêmeas; d) amplitude de variação do número de ovos postos por fêmea em cada data de oviposição; e) variação temporal das posturas em relação à frequência de fêmeas que ovipositam em cada data; e f) quantidade de ovos postos ao longo do tempo. Trata-se apenas de uma análise descritiva sem ajuste de modelos para representar tais padrões. Para tanto, seria necessário o ajuste de modelos não lineares que considerassem possíveis autocorrelações entre os números de ovos postos em cada data por uma mesma fêmea.

Os comandos para construção dos gráficos dos padrões de oviposição estão descritos na Figura 3. Pode ser incluído um título para o gráfico, utilizando-se os comandos `"title "Padrão temporal de oviposição";"`. Neste programa, o comando para inclusão do título está desativado⁴. Para construir um gráfico para cada grupo basta adicionar os comandos `"by GRUPO;"` e retirar a opção `"=GRUPO"` em `plot N_OVOS*IDADE=grupo/` `haxis=axis1 vaxis=axis2;"` (Fig. 3). O tipo e tamanho da fonte a serem utilizados nos gráficos são definidos em `goptions` pelas opções `ftext='Univers/bold'` e `htext=4.5` respectivamente (Fig. 3 e Fig. 4). Vários tipos de fontes podem ser utilizados nos gráficos. Para detalhes ver o manual "The GPLOT procedure" (SAS, 1990). O gráfico resultante é apresentado na Figura 4.

```

/* ***** */
/* Construção de gráficos do padrão de oviposição em cada GRUPO */
/* ***** */
/* Ordenação dos dados por grupo e idade */
data FERT1;
set DATA1;
keep GRUPO N_FEMEA IDADE N_OVOS;
proc sort data=FERT1 out=FERT2;
by GRUPO IDADE;

proc means data=FERT2 noprint;
var IDADE N_OVOS;
Output out=FERT3 min=IDADEMIN OVOSMIN max=IDADEMAX OVOSMAX;

/* ***** */
/* Definição dos valores máximo e mínimo para as escalas dos eixos x e y do gráfico */
/* de oviposição */
/* ***** */
data _null_;
set FERT3;
IDADEMIN=IDADEMIN-0.5;
IDADEMAX=IDADEMAX+0.5;
call symput('IDADEMIN',IDADEMIN);
call symput('IDADEMAX',IDADEMAX);
call symput('OVOSMIN',OVOSMIN);
call symput('OVOSMAX',OVOSMAX);

data FERT2;
set FERT2;
where N_OVOS>0;

label IDADE= 'Idade da fêmea (dias)';
label GRUPO= 'Grupo';
label N_OVOS= 'Número de ovos';
run;

```

Fig. 3. Comandos do programa SAS para construção dos gráficos do padrão de oviposição em cada grupo.

```

/* Formatação do gráfico */
goptions reset=global unit=pct
ftext='Univers/bold' htext=4.5;
/* title "Padrão temporal de oviposição"; */
/* Grupo 1 */
symbol interpol=none /* tipo de interpolação */
cv=black /* cor do símbolo */
width=1 /* espessura da linha */
value=dot /* tipo de símbolo */
height=2; /* tamanho do símbolo */
/* Grupo 2 */
symbol2 cv=red value=dot height=2;
/* Grupo 3 */
symbol3 cv=blue value=dot height=2;


/* Formatação do eixo x */
axis1 label=(angle=0 h=4.5) /* ângulo entre o rótulo e o eixo x
e tamanho da fonte */
offset=(1,1) /* distancia entre a origem do eixo x e o primeiro marcador
e entre o fim do eixo x e o ultimo marcador */
order=&IDADEMIN to &IDADEMAX BY 1 /* escala do eixo x */
length=90; /* Comprimento do eixo x */

/* Formatação do eixo y */
axis2 label=(angle=90 h=4.5) /* ângulo entre o rótulo e o eixo x
e tamanho da fonte */
order=&OVOSMIN to &OVOSMAX BY 10 length=60; /* escala do eixo y */ run;

/* Construção do gráfico de acordo com a formatação pré-definida */
proc gplot data=FERT2;
/* title "Padrão temporal de oviposição */
plot N_OVOS*IDADE=GRUPO /* comando para construção do gráfico */

haxis=axis1 vaxis=axis2; /* definição dos eixos */
run;

```

Fig. 4. Comandos do programa SAS para construção dos gráficos do padrão de oviposição em cada grupo (Continuação).**Fig. 5.** Padrões de oviposição das fêmeas durante o período de avaliação nos três grupos.

⁴ Qualquer parte de um programa SAS pode ser temporariamente inativada escrevendo-se os respectivos comandos entre os símbolos `/*` e `*/`. Os comandos serão então interpretados como comentários. Esse recurso é útil para suprimir ações que não são de interesse do usuário num determinado momento, mas que podem ser necessárias futuramente.

Cálculo do número total de ovos postos por fêmea durante o período de oviposição

O número total de ovos postos por fêmea durante o período de oviposição é um indicador de prolificidade. Nesta etapa, somam-se os números de ovos de cada fêmea em todas as posturas, utilizando o PROC MEANS (Fig. 6). Os dados gerados neste procedimento são listados na janela "output" (Fig. 7) e exportados para um arquivo Excel (oviposicao exemplo.xls) que poderá ser utilizados para realização de testes estatísticos para comparação de grupos quanto ao número médio de ovos postos por fêmea. Esse tipo de análise não está incluído neste programa. Recomenda-se o uso de modelos lineares generalizados que consideram a distribuição de Poisson para modelar a variável número de ovos (Hinde & Demetrio, 1998). Quando há um grande número de zeros, a questão da sobredispersão deve ser considerada.

Para gravação de dados gerados pelo programa SAS num arquivo externo (Excel 2000, por exemplo) utilizamos o **PROC EXPORT**. Em data=FERT, informamos o nome do arquivo interno SAS a ser exportado, em outfile="C:\nome do diretório\oviposicao.xls", o nome do arquivo Excel 2000 onde os dados de oviposição serão armazenados (Anexo 3) Esses dados são utilizados para construção de gráficos de caixas (box-plots) para o número médio de ovos postos por fêmea durante sua vida, utilizando o **PROC GLOT** (Fig. 8).

```

/****** */
/*Cálculo do número total de ovos postos por fêmea durante o período de */
/*oviposição */
/****** */
proc sort data=DATA1;
by GRUPO N_FEMEA IDADE;

proc means data=DATA1 sum noprint;
by GRUPO N_FEMEA;
var N_OVOS;
output out=FERT4 sum=OVTOTAL;

proc print data=FERT4 noobs;
title 'No.de ovos por fêmea durante o período total de oviposição';
by GRUPO;
var N_FEMEA OVTOTAL;
run;

/****** */
/*Exportação dos dados de numero de ovos postos por fêmea durante o período */
/*total de oviposição para um arquivo Excel2000 (oviposicao exemplo.xls) */
/****** */
proc export data=FERT4
outfile="C:\nome do diretório\oviposicao exemplo.xls"
DBMS=EXCEL2000 replace;
run;

```

Fig. 6. Código SAS para cálculo do número total de ovos postos por fêmea em cada grupo e exportação dos dados resultantes para um arquivo Excel 2000.

Nº de ovos por fêmea durante o período total de oviposição

grupo 1	
n_femea	OVTOTAL
1	49
2	65
3	41
4	40
5	41
6	68
7	65
8	42
9	54
10	54

grupo 2	
n_femea	OVTOTAL
1	97
2	119
3	110
4	137
5	127
6	87
7	100
8	107
9	77

grupo 3	
n_femea	OVTOTAL
1	133
2	117
3	87
4	205
5	119
6	203
7	133

Fig. 7. Listagem do número de ovos por fêmea durante o período de avaliação em cada grupo.

```

/****** */
/*Construção de gráficos de caixas para o número de ovos postos por fêmea em */
/*cada GRUPO */
/****** */
data FERT4;
set FERT4;
label OVTOTAL= 'No.de ovos/fêmea';
label GRUPO= 'GRUPO';

options reset=global gunit=pct
ftext='Univers/bold' htext=4.5;

/*title "Box plot do numero de ovos por fêmea";*/
symbol interpol=boxtf05 /* tipo de box plot */
cv=red /* cor do interior da "caixa" */
co=black /* cor dos limites da "caixa" */
width=2 /* espessura da linha */
value=dot /* tipo de simbolo */
height=2; /* tamanho do simbolo */

axis1 offset=(5,5)length=25;

axis2 label=(angle=90)length=35;
proc gplot data=FERT4;
plot OVTOTAL*GRUPO/haxis=axis1 vaxis=axis2;
run;

```

Fig. 8. Comando SAS para construção de gráficos de caixas do número de ovos por fêmea durante o período de avaliação para cada grupo.

Construção de gráficos de caixa para o número de ovos postos por fêmea durante o período de avaliação.

Os gráficos de caixas mostram a variabilidade entre fêmeas de um mesmo grupo. Neste exemplo, as caixas são limitadas pelos percentis 5 e 95 da distribuição do número de ovos por fêmea. Outros percentis podem ser escolhidos, modificando a opção "interpol" (Fig. 8). Quanto maior a altura da caixa (distância entre os percentis 5 e 95) maior a variabilidade dos dados. A visualização conjunta das caixas correspondentes a cada grupo permite avaliar a homogeneidade de variâncias, pressuposto importante dos modelos lineares (não generalizados). A barra horizontal, dentro da caixa, representa a mediana da variável em questão; a distância entre mediana e a altura média da caixa é um indicador da assimetria da distribuição.




Fig. 9. Gráfico de caixas do número de ovos por fêmea durante o período de avaliação em cada grupo. Os limites inferior e superior da caixa correspondem aos percentis 5 (p_5) e 95 (p_{95}), respectivamente, da distribuição empírica do no. de ovos por fêmea. As linhas verticais unem os valores mínimo e máximo observados em cada grupo (até uma distância equivalente a 1,5 vezes a distância entre p_{25} e p_{75}). Valores inferiores ou superiores aos limites da linha vertical são representados como pontos isolados ("outliers").

Observa-se que no grupo 3 (Fig. 9) as fêmeas apresentam maior variabilidade quanto ao número de ovos; a distribuição empírica para este grupo é bastante assimétrica, com valores de 142,43 e 133 ovos por fêmea para a média e mediana amostrais respectivamente. Os dados evidenciam uma dependência entre média e variância, padrão coerente com a distribuição de Poisson considerada na geração dos dados.

Geração de pseudo amostras para uso no método Jackknife.

Para aplicação do método Jackknife, é necessário gerar NG conjuntos de dados com NG-1 fêmeas (uma fêmea retirada em cada subconjunto). Os procedimentos para estimação dos parâmetros serão repetidos em cada subconjunto ($V = 1$ a NG) e também para o conjunto de dados originais (iteração $V = 0$). A obtenção de estimativas diferentes do mesmo parâmetro permite quantificar a variância associada a essas estimativas. Nas Figuras 10 a 12 apresentamos os códigos SAS utilizados em cada passo para geração desses subconjuntos de dados.

```

/******
/* Cálculo do número de indivíduos (fêmeas) em cada GRUPO
/******
proc sort data=DATA1 nodupkey out=DATA2;
by GRUPO N_FEMEA;

proc means noprint data=DATA2;
var N_OVOS;
by GRUPO;
output out=DATA3 n=NINDGR;

proc means noprint data=DATA3;
var NINDGR;
output out=DATA4 max=MAXNGR;

proc print data=DATA3 noobs;
title "Número de indivíduos por grupo";
var GRUPO NINDGR;
run;

```

Fig. 10 Cálculo do número de indivíduos (fêmeas) em cada grupo.

Como os números de identificação dos indivíduos em cada grupo nem sempre estão em ordem sequencial, o número do último indivíduo nem sempre corresponde ao número de indivíduos no grupo. Assim, utilizamos a opção $n=NINDGR$ no "output" do PROC MEANS para gerar um arquivo com o número de indivíduos por grupo. A seguir é criado um arquivo onde os números de identificação dos indivíduos (INDNUM) estão em ordem sequencial (Fig. 11).

```

/******
/* Criação de um arquivo com números de identificação de cada fêmea em ordem
/* sequencial
/******
data null ;
set DATA4;
call symput('N',MAXNGR);

data T;
set DATA3;
do INDNUM=1 to &N;
if INDNUM>NINDGR then delete;
drop _type_ _freq_;
output;
end;

data T;
merge T DATA2;
by GRUPO;

data DATA1SEQ;
merge DATA1 T;
by GRUPO N_FEMEA;

proc sort data=DATA1SEQ;
by INDNUM;
/*proc print data=DATA1SEQ;*/
run;

```

Fig. 11. Código SAS para criação de um arquivo com números de identificação de cada fêmea em ordem sequencial.

Assim, o valor máximo do INDNUM em cada grupo corresponde ao número de indivíduos naquele grupo (NG). São gerados então os NG subconjuntos com (NG-1) indivíduos (Fig. 10). O arquivo que contém os dados dos NG subconjuntos é anexado ao conjunto de dados original ($V = 0$). O conjunto de dados original será utilizado para estimar os parâmetros utilizando o método aproximado. Essas estimativas serão os valores iniciais necessários para implementar os procedimentos iterativos em cada subconjunto ($V = 1$ a NG).

Nesta etapa, é utilizada uma MACRO⁵ do SAS[®] para criação de um conjunto de dados que contém os dados originais ($V = 0$) e com os NG subconjuntos com (NG-1) fêmeas.

⁵O recurso Macro é uma ferramenta que permite usar toda a capacidade do Sistema SAS e criar nele rotinas dedicadas, ou seja, adaptadas às necessidades específicas de cada usuário. Isso permite abreviar ou substituir, de maneira fácil e conveniente, uma grande quantidade de linhas de programação. Esse recurso usa uma linguagem de programação que permite executar partes de um programa SAS de forma condicionada; construir ambientes para entrada de dados por parte dos usuários; e ainda é uma maneira de compartilhar informações entre etapas de um programa.

Quando se escreve um trecho do programa na linguagem Macro, é dado a ele um nome, pelo qual será ativado o trecho sempre que necessário no programa. No programa aqui apresentado, foi escrita uma Macro, chamada de ALI, que tem por objetivo criar um novo conjunto de dados, a partir dos dados originais. O uso da Macro ALI é útil nesse caso, pois o novo conjunto é dependente de informações que são produzidas em outras etapas do programa. Além disso, sem a utilização desse recurso, um número muito maior de comandos seria necessário para realizar a mesma tarefa.

```

/******
/* Criação de um arquivo (DATA6) com NG + 1 subconjuntos (V), sendo o primeiro
/* (V = 0) com todas as NG fêmeas em cada um dos respectivos
/* tratamentos e os outros NG subconjuntos com NG-1 fêmeas. Para V = 1, retira-se
/* a fêmea 1 e assim sucessivamente até a fêmea NG. Este conjunto de dados será
/* utilizado para estimar os parâmetros da TVF utilizando o método jackknife.
/******
data DATA6;
set DATA1SEQ;
V = 0;
%macro ALI;
%do I = 1 %to &N;
%let G = &I;

data DATA5;
set DATA1SEQ;
V = &G;
if INDNUM = V then delete;

data DATA6;
merge DATA6 DATA5;
by V;
%end;
%mend ALI;
%ALI
run;

```

Fig. 12. Comandos SAS para criação de um conjunto de dados que contém os dados originais (V=0) e com os NG subconjuntos com (NG-1) fêmeas. Em cada subconjunto V foram retirados os dados da fêmea V.

Cálculo do número médio de ovos (fêmeas) postos por fêmea (MX) em cada data de oviposição

A taxa líquida de reprodução (R_0) em cada grupo é calculada somando-se, para todo o período de avaliação, o número médio de ovos do sexo feminino postos por fêmea (MX) em cada data de oviposição. Para calcular MX multiplica-se o número de ovos postos por fêmea, em cada data, pela porcentagem de fêmeas na população (PCT_FEM) (Fig. 13).

Os resultados gerados para cada grupo são armazenados no arquivo DATA7 que contém as seguintes variáveis: grupo, subconjunto (V), idade da fêmea (IDADE), número de indivíduos (fêmeas) no grupo (NINDGR), sobrevivência na fase imatura (SFIMAT) e número de descendentes fêmeas produzidas por fêmea para a geração seguinte (MX). Para listar o arquivo DATA7 na janela "output" utiliza-se o comando PROC PRINT.

Estimação dos parâmetros associados às TBVF para o subconjunto com os dados originais utilizando o método aproximado

As tabelas de vida e fertilidade condensam informações sobre reprodução e sobrevivência das populações de fêmeas, em cada grupo. São constituídas das seguintes colunas: a) tratamento ou grupo (GRUPO); b) idade da fêmea (IDADE); c) índice de sobrevivência acumulado até a idade X (LX); d) número médio de fêmeas

produzidas por fêmea em cada data de oviposição (MX); e) número médio líquido de fêmeas produzidas por fêmea em cada data de oviposição (LXMX) e f) o produto $X * LX * MX$ que é utilizado para calcular uma estimativa aproximada do intervalo entre gerações. Os comandos SAS para construção de TBVF para cada grupo G são apresentados na Figura 14.

```

/******
/* Cálculo do número médio de ovos postos por fêmea (NM_OVFEM) em cada data
/* de oviposição
/******
/* Ordenação dos dados
proc sort data = DATA6;
by GRUPO V IDNUM IDADE;

data DATA6;
set DATA6;
if V > NINDGR then delete;

proc sort data = DATA6;
by GRUPO V IDADE;

/* Cálculo do no. de ovos do sexo feminino, por fêmea, em cada data de oviposição */
data DATA6;
set DATA6;
NM_OVFEM = N_OVOS * PCT_FEM;

proc means data = DATA6 noprint;
id NINDGR SFIMAT;
by GRUPO V IDADE;
var NM_OVFEM;
output out = DATA7 mean = MX;
run

```

Fig. 13. Cálculo do número médio de ovos postos por fêmea (NM_OVFEM) em cada data de oviposição e da contribuição de cada data de oviposição (MX) para o crescimento da população de fêmeas.

```

/******
/* Construção de tabelas de vida e fertilidade com os dados de cada subconjunto V
/******
data DATA8;
set DATA7;
N_FVIVAS = _freq_;
if V = 0 then LX = SFIMAT * (N_FVIVAS / NINDGR);
if V > 0 then LX = SFIMAT * (N_FVIVAS / (NINDGR - 1));
LXMX = LX * MX;
MXLXX = LXMX * IDADE;
drop _type_ _freq_;
run;

/* Exportação para um arquivo Excel2000, dos dados das tabelas de vida e
/* fertilidade de cada grupo, construídas com os dados originais (V = 0)
/******
data DATA8V0;
set DATA8;
where V = 0;
keep GRUPO IDADE LX MX LXMX MXLXX;

proc export data = DATA8V0
outfile = "C:\ nome do diretorio \tabela de vida exemplo.xls"
DBMS = EXCEL2000 replace;
run;

proc print data = DATA8V0 noobs;
title 'Tabela de vida e fertilidade';
by GRUPO;
var IDADE LX MX LXMX MXLXX;
run;

```

Fig. 14. Códigos SAS para construção das tabelas de vida e fertilidade para cada grupo considerando apenas os dados originais (subconjunto V=0).

A tabela de vida construída com os dados originais (DATA8V0) é exportada para o arquivo Excel 2000 "TabelaDeVidaExemplo.xls" (Anexo 4) e listado na janela "output" (Fig. 15).

Os parâmetros associados às TBVF em cada grupo são então estimados utilizando o método aproximado. Os códigos SAS utilizados nesse processo são apresentados na Figura 16. Nesse método o intervalo entre gerações para uma particular fêmea é estimado como uma média do tempo entre a postura do ovo que originou uma fêmea e a data de postura dos ovos que originarão os seus descendentes, ponderada pelo número líquido de

descendentes do sexo feminino produzido em cada data de postura. Essa estimativa é obtida dividindo-se $LX.MX$ pelo produto $X.LX.MX$. As estimativas obtidas quando se utiliza o método aproximado são listadas na tela "output" (Fig. 17).

grupo 1				
idade	LX	MX	LXMX	MXLXX
16.5	0.7	1.519	1.0633	17.5445
17.5	0.7	3.626	2.5382	44.4185
18.5	0.7	3.528	2.4696	45.6876
19.5	0.7	3.381	2.3667	46.1507
20.5	0.7	4.851	3.3957	69.6119
21.5	0.7	3.038	2.1266	45.7219
22.5	0.7	1.519	1.0633	23.9243
23.5	0.7	1.764	1.2348	29.0178
24.5	0.7	2.107	1.4749	36.1351
25.5	0.7	0.098	0.0686	1.7493

grupo 2				
idade	LX	MX	LXMX	MXLXX
16.5	0.8	3.38889	2.71111	44.733
17.5	0.8	4.77778	3.82222	66.889
18.5	0.8	8.22222	6.57778	121.689
19.5	0.8	6.55556	5.24444	102.267
20.5	0.8	7.33333	5.86667	120.267
21.5	0.8	7.00000	5.60000	120.400
22.5	0.8	9.11111	7.28889	164.000
23.5	0.8	3.38889	2.71111	63.711
24.5	0.8	3.61111	2.88889	70.778
25.5	0.8	0.00000	0.00000	0.000

grupo 3				
idade	LX	MX	LXMX	MXLXX
16.5	0.9	5.2000	4.6800	77.220
17.5	0.9	7.8743	7.0869	124.020
18.5	0.9	10.1029	9.0926	168.213
19.5	0.9	7.5771	6.8194	132.979
20.5	0.9	10.6971	9.6274	197.362
21.5	0.9	13.0743	11.7669	252.987
22.5	0.9	10.4743	9.4269	212.104
23.5	0.9	5.1257	4.6131	108.409
24.5	0.9	3.6400	3.2760	80.26
25.5	0.9	0.2971	0.2674	6.819

Fig. 15. Tabelas de vida e fertilidade para cada grupo.

```

/*-----*/
/*Cálculo da taxa líquida de reprodução (Ro, no. fem/fem) e do numerador da
/*expressão IMG = (soma(MX*LX*X))/Ro para estimativa do intervalo entre
/*gerações (IMG) pelo método aproximado.
/*-----*/
proc means noprint data=DATA8;
id NINDGR;
by GRUPO V;
var LXMX MXLXX;
output out=DATA9 sum=RO NUMT;
run;

/*Estimativa dos seguintes parâmetros utilizando o método aproximado:
/*Ro: taxa líquida de reprodução
/*IMG : intervalo médio entre gerações
/*TD: tempo de duplicação
/*Rm: taxa intrínseca de crescimento
/*Lambda: razão finita de crescimento
/*-----*/

data DATA10;
set DATA9;
IMG = NUMT/RO;
RM = (LOG(RO))/IMG;
TD = (LOG(2))/RM;
LAMBDA = exp(RM);
keep GRUPO NINDGR V RO IMG TD RM LAMBDA;
run;

Impressão na tela "output" das estimativas dos parâmetros utilizando método
aproximado
proc print data=DATA10 noobs;
title 'Estimativas dos parâmetros obtidas pelo o método aproximado';
where V=0;
var GRUPO RO IMG TD RM LAMBDA;
run;

```

Fig. 16. Códigos SAS para estimação dos parâmetros associados às TBVF de cada um dos grupos, utilizando o método aproximado.

Estimativas dos parâmetros obtidas pelo o método aproximado					
grupo	RO	IMG	TD	RM	LAMBDA
1	17.8017	20.2206	4.86781	0.14239	1.15303
2	42.7111	20.4802	3.78105	0.18332	1.20120
3	66.6566	20.4087	3.36851	0.20577	1.22847

Fig. 17. Estimativas dos parâmetros associados às TBVF, obtidas pelo o método aproximado.

Estimação dos parâmetros utilizando métodos iterativos

A taxa líquida de reprodução (Ro) é estimada, em cada grupo G e subconjunto V , pela somatória dos valores de $LX.MX$ para todo o período de avaliação ($ro = \sum LX.MX$), onde $LX.MX$ corresponde à contribuição líquida de cada data de postura para a geração seguinte, em termos de número de descendentes gerados por fêmea. Convencionamos utilizar símbolos iniciando com letras maiúsculas, para representar os parâmetros, e símbolos com todas as letras minúsculas, para representar seus respectivos estimadores.

As estimativas do parâmetro Rm obtidas no item anterior são utilizadas nas etapas seguintes onde os parâmetros associados às TBVF em cada grupo e cada subconjunto V , são estimados através de um processo iterativo. Esse processo busca encontrar, em cada grupo G e subconjunto V , o valor de Rm para o qual a somatória dos termos $X.LX.exp(Rm.X)$ é igual à unidade. Essa expressão é derivada de um modelo de crescimento exponencial onde se pressupõe que a cada intervalo de tempo unitário a população é multiplicada por um fator constante $\lambda = exp(Rm)$.

A busca do valor do Rm em cada grupo é feita no intervalo entre $0,5*rm_o$ e $1,5*rm_o$ a intervalos constantes de 0,005, onde rm_o é a estimativa do parâmetro Rm obtida pelo método aproximado. O processo resulta em estimativas de Rm para cada um dos NG subconjuntos de $(NG-1)$ fêmeas. As estimativas dos demais parâmetros são calculadas para cada subconjunto V , em função das estimativas dos parâmetros Rm e Ro , utilizando as seguintes expressões: a) $td = \log_2(rm)$; b) $img = \log(ro)/rm$ e c) $l = exp(rm)$ (Fig. 18 e Fig. 19). As estimativas dos parâmetros oriundas do conjunto de dados originais são listadas na tela "output" (Fig. 20).

Cálculo dos pseudo-valores correspondentes aos parâmetros Ro , Rm , IMG , TD , e $Lambda$

As estimativas dos parâmetros obtidas em cada subconjunto V são então corrigidas utilizando-se a expressão $RmPSV_v = NG*(Rm_o) - (NG-1)*Rm_v$, onde V varia de 1 a NG e $RmPSV_v$ é denominado pseudovalor de Rm correspondente ao subconjunto V (Figura 15). O mesmo procedimento é utilizado para obtenção dos pseudovalores de Ro , IMG , Dt e $Lambda$ (Fig. 21). Os NG pseudovalores em cada grupo são exportadas para um arquivo Excel 2000, pseudovalores Exemplo.xls (Anexo 5).

```

/*Estimação dos parâmetros utilizando métodos iterativos */
data DATA11;
merge DATA8 DATA10;
by GRUPO V;
keep GRUPO NINDGR V IDADE RM RO LXXM;

data DATA12;
set DATA11;
do U = 500 to 1500 by 5;
R = RM*(U/1000);
Y = LXXM*(exp(-R*IDADE));
keep GRUPO NINDGR V IDADE Y U R RO;
output;
end;

proc sort data = DATA12;
by GRUPO V R;

proc means noprint data = DATA12;
by GRUPO V R;
var Y;
id U RO NINDGR;
output out = DATA13 sum = SUM;
data DATA14;
set DATA13;
DELTA = abs(1-SUM);
drop _type_ _freq_;

proc means noprint data = DATA14;
by GRUPO V;
var DELTA;
output out = MINDELTA min = MINDELTA;
run;

```

Fig. 18. Estimação dos parâmetros associados às TBVF, em cada grupo e subconjunto V, utilizando o processo iterativo descrito nesta seção.

```

/*Estimação dos parâmetros utilizando métodos iterativos */
data DATA15a;
merge DATA14 MINDELTA;
by GRUPO V;

data DATA15;
set DATA15a;
where DELTA = MINDELTA;
RM = R;
TD = (log(2))/RM;
IMG = (log(RO))/RM;
LAMBDA = exp(RM);
keep GRUPO NINDGR V SUM MINDELTA RO RM TD IMG LAMBDA;
run;

data DATA16;
set DATA15;
where V = 0;
ROO = RO;
RMO = RM;
IMGO = IMG;
TDO = TD;
LAMBDAO = LAMBDA;
keep GRUPO NINDGR ROO RMO IMGO TDO LAMBDAO;
run;

proc print data = DATA16 noobs;
title 'Estimativas dos parâmetros utilizando o conjunto de dados original';
by GRUPO;
var ROO RMO IMGO TDO LAMBDAO;
run;

data DATA17;
set DATA15;
where V > 0;
keep GRUPO NINDGR V RO RM IMG TD LAMBDA;
proc sort data = DATA17;
by GRUPO;
proc sort data = DATA17;
by GRUPO;

data DATA18;
merge DATA16 DATA17;
by GRUPO;
run;

```

Fig. 19. Estimação dos parâmetros associados às TBVF em cada grupo e subconjunto V, utilizando o processo iterativo descrito nesta seção (Continuação).

Estimativas dos parâmetros utilizando o conjunto de dados original				
grupo = 1				
ROO	RMO	IMGO	TDO	LAMBDAO
17.8017	0.14524	19.8241	4.77237	1.15632
grupo = 2				
ROO	RMO	IMGO	TDO	LAMBDAO
42.7111	0.18790	19.9807	3.68883	1.20672
grupo = 3				
ROO	RMO	IMGO	TDO	LAMBDAO
66.6566	0.21092	19.9110	3.28635	1.23481

Fig. 20. Estimativas dos parâmetros associados às TBVF, obtidas utilizando o conjunto de dados originais.

```

/*Cálculo dos pseudo-valores para RO(ROPSV), RM(RMPSV), IMG(IMGPSV)
*DT(TDPSV) e LAMBDA(LPSV) utilizando os respectivos valores estimados para
*cada subconjunto V e os valores obtidos do conjunto de
*dados original
*/
data DATA19;
set DATA18;
ROPSV = (NINDGR*ROO)-((NINDGR-1)*RO);
RMPSV = (NINDGR*RMO)-((NINDGR-1)*RM);
IMGPSV = (NINDGR*IMGO)-((NINDGR-1)*IMG);
TDPSV = (NINDGR*TDO)-((NINDGR-1)*TD);
LPSV = (NINDGR*LAMBDAO)-((NINDGR-1)*LAMBDA);
proc print data = DATA19 noobs;
title 'Pseudo-valores de Ro';
var GRUPO ROPSV;
run;

data DATA19;
set DATA18;
ROPSV = (NINDGR*ROO)-((NINDGR-1)*RO);
RMPSV = (NINDGR*RMO)-((NINDGR-1)*RM);
IMGPSV = (NINDGR*IMGO)-((NINDGR-1)*IMG);
TDPSV = (NINDGR*TDO)-((NINDGR-1)*TD);
LPSV = (NINDGR*LAMBDAO)-((NINDGR-1)*LAMBDA);
proc print data = DATA19 noobs;
title 'Pseudo-valores de Ro';
var GRUPO ROPSV;
run;

data PSEUDO;
set DATA19;
keep GRUPO V ROPSV RMPSV TDPSV IMGPSV LPSV;
run;

/*Exportação do arquivo de dados (PSEUDO) com os NG pseudo-valores de Ro, Rm,
*IMG, Dt, e Lambda em cada GRUPO
*/
data PSEUDO;
set DATA19;
keep GRUPO V ROPSV RMPSV TDPSV IMGPSV LPSV;
proc export data = PSEUDO
outfile = "C:\nome do diretorio\pseudovalores.xls"
DBMS = EXCEL2000 replace;
run;

```

Fig. 21. Códigos SAS para cálculo dos pseudovalores de Ro, Rm, IMG, Dt e Lambda e exportação dos resultados para um arquivo Excel.

Quando os grupos apresentarem variâncias homogêneas, análise de variância e procedimentos de comparação múltipla ou análises de contrastes associados podem ser realizados utilizando os pseudovalores.

Construção de gráficos de caixas para os parâmetro associados às TBVF em cada grupo.

Os gráficos de caixas apresentados nesta seção (Fig. 23) foram construídos para visualização de algumas características das distribuições dos pseudovalores, tais como média, variância e assimetria, conforme discutido na seção 3.6 onde o mesmo tipo de gráfico é apresentado para a variável número de ovos por fêmea. Detalhes do código SAS são apresentados na Figura 22.

Quanto mais simétricas forem as distribuições dos pseudovalores, mais adequado será o uso dos testes baseados na distribuição t utilizando estimativas Jackknife da variância. Distribuições extremamente assimétricas combinadas com tamanhos de amostra pequenos podem comprometer o uso deste método.

Estimação de médias e variâncias dos parâmetros associados às TBVF em cada grupo

As estimativas Jackknife das médias e variâncias dos parâmetros associados às TBVF em cada grupo são obtidas calculando-se as médias e variâncias amostrais dos respectivos pseudovalores, utilizando o PROC MEANS (Fig. 23).

```

/******
/*Construção de gráficos de caixas para cada parâmetro associado à da tabela de
/*vida em cada GRUPO.
/******

data DATA19;
set DATA19;
label ROPSV = 'Ro (fêmea/fêmea)';
label RMPVS = 'Rm';
label IMGPSV = 'IMG (dias)';
label TDPVS = 'Dt (dias)';
label LPSV = 'Lambda';

label GRUPO = 'Grupo';

options reset = global gunit = pct
ftext = 'Univers/bold' htext = 4.5;

symbol interpol = boxtf05 /* tipo de box plot */
cv = red /* cor do interior da "caixa" */
co = black /* cor dos limites da "caixa" */
width = 2 /* espessura da linha */
value = dot /* tipo de símbolo */
height = 2; /* tamanho do símbolo */

axis1 offset = (5,5) minor = none length = 25;

axis2 label = (angle = 90) minor = none major = (n = 5) length = 35;

proc gplot data = DATA19;
plot (ROPSV RMPVS IMGPSV TDPVS LPSV) *GRUPO/haxis = axis1
vaxis = axis2;
run;
quit;

```

Fig. 22. Códigos SAS para construção de gráficos de caixas para cada parâmetro associado às TBVF.




Fig. 23. Gráficos de caixas construídos com os pseudovalores das estimativas dos parâmetros taxa líquida de reprodução (Ro), taxa intrínseca de crescimento (Rm), tempo de duplicação (IMG), intervalo entre gerações (Dt) e razão finita de crescimento (Lambda) para os grupos 1, 2 e 3.

```

/******
/*Criação de um arquivo de dados (DATA20) com médias e variâncias dos pseudo
/*valores de Ro, Rm, IMG, Dt, e Lambda.
/******

proc means noprint data = DATA19;
id NINDGR;
by GRUPO;
where V > 0;
var ROPSV RMPVS IMGPSV TDPVS LPSV;
output out = DATA20 mean = RO_MEDIA RM_MEDIA IMG_MEDIA DT_MEDIA
L_MEDIA
var = RO_VAR RM_VAR IMG_VAR DT_VAR L_VAR;
run;

```

Fig. 24. Códigos SAS para estimação das médias e variâncias dos parâmetros associados às TBVF.

Criação de um arquivo de dados para comparação dos parâmetros entre pares de grupos

Nesta etapa são realizadas operações de rearranjo nas linhas e colunas das matrizes de dados de modo a permitir comparações entre todos os possíveis pares de tratamentos (Fig. 19 a 21). Esse tipo de análise é adequada para tratamentos qualitativos, não estruturados e com variâncias heterogêneas, onde contrastes entre qualquer par de tratamentos são de interesse.

Nos casos em que os tratamentos são qualitativos, com variâncias homogêneas, testes de estatísticos baseados em estimativas de variância derivadas do quadrado médio do resíduo (Duncan, Tukey, teste F para contrastes) da análise de variância dos pseudovalores podem ser utilizados. Quando os tratamentos são quantitativos, o uso de modelos de regressão lineares ou não lineares é recomendável para quantificar o efeito tratamento sobre os parâmetros associados às TBVF (Zar, 1984). Esses métodos não foram incluídos neste trabalho.

```

/******
/*Criação de um arquivo de dados (T_DATA5) com todas as comparações entre pares de
/*grupos. Esses dados serão utilizados para testes t considerando variâncias não homogêneas.
/*
/******

data _null_;
set DATA20 nobs=NGR;
call symput('NGR',NGR);

data T_DATA1;
set DATA20;
GR=_n_;
GR1=GRUPO;
GR2=GRUPO;
GR_A=GR;
GR_B=GR;

keep GRUPO GR GR_A GR_B GR1 GR2 NINDGR
RO_MEDIA RM_MEDIA IMG_MEDIA TD_MEDIA L_MEDIA
RO_VAR RM_VAR IMG_VAR TD_VAR L_VAR;

data T_DATA1A;
set T_DATA1;
if GR<=NGR;
drop GR2 GR_B;

data T_DATA1B;
set T_DATA1;
if GR>1;
drop GR1 GR_A;

data T_DATA2;
F=&NGR-1;
do GR1=1 to F;
Z=GR1+1;
do GR2=Z to &NGR;
output;
end;
end;

```

Fig. 25. Criação de um arquivo de dados com para comparação dos parâmetros entre pares de grupos.

Teste t de Student para comparação entre pares de tratamentos.

Para comparação de um par de tratamentos (grupos) com relação a um parâmetro de interesse, primeiro são calculadas diferenças entre as estimativas do parâmetro em cada grupo e os correspondentes erros padrão das diferenças, considerando variâncias heterogêneas. As estatísticas t são obtidas dividindo-se o valor de cada diferença pelo respectivo erro padrão; os níveis de significância nominais (valores p) associados a cada valor de t são calculados utilizando a função PROBT, para os testes bilateral, unilateral à esquerda e unilateral à direita (Fig. 28).

```

data T_DATA3;
merge T_DATA2 T_DATA1A;
  by GR1;
  RO_MED1A1=RO_MED1A;
  RM_MED1A1=RM_MED1A;
  IMG_MED1A1=IMG_MED1A;
  TD_MED1A1=TD_MED1A;
  L_MED1A1=L_MED1A;
  RO_VAR_A=RO_VAR;
  RM_VAR_A=RM_VAR;
  IMG_VAR_A=IMG_VAR;
  TD_VAR_A=TD_VAR;
  L_VAR_A=L_VAR;
  NINDGR1=NINDGR;
  keep GR1 GR2 GR_A NINDGR1 RO_MED1A1 RM_MED1A1 IMG_MED1A1
  TD_MED1A1 L_MED1A1 RO_VAR_A RM_VAR_A IMG_VAR_A TD_VAR_A L_VAR_A;
proc sort data=T_DATA3;
  by GR2 GR1;
data T_DATA4;
merge T_DATA3 T_DATA1B;
  by GR2;
  RO_MED2A=RO_MED2;
  RM_MED2A=RM_MED2;
  IMG_MED2A=IMG_MED2;
  TD_MED2A=TD_MED2;
  L_MED2A=L_MED2;
  RO_VAR_B=RO_VAR;
  RM_VAR_B=RM_VAR;
  IMG_VAR_B=IMG_VAR;
  TD_VAR_B=TD_VAR;
  L_VAR_B=L_VAR;
  NINDGR2=NINDGR;
  keep GR1 GR2 GR_A GR_B
  NINDGR1 NINDGR2
  RO_MED1A1 RM_MED1A1 IMG_MED1A1 TD_MED1A1 L_MED1A1
  RO_VAR_A RM_VAR_A IMG_VAR_A TD_VAR_A L_VAR_A
  RO_MED2A RM_MED2A IMG_MED2A TD_MED2A L_MED2A
  RO_VAR_B RM_VAR_B IMG_VAR_B TD_VAR_B L_VAR_B;
proc sort data=T_DATA4;
  by GR1 GR2;
data T_DATA4;
set T_DATA4;
  N1=NINDGR1;
  N2=NINDGR2;
run;

```

Fig. 26. Códigos SAS para criação de um arquivo de dados para comparação dos parâmetros entre pares de grupos (Continuação).

```

data T_DATAS;
set T_DATA4;

PARAM='RO'; MED_A=RO_MED1A1; MED_B=RO_MED2A; VAR_A=RO_VAR_A;
VAR_B=RO_VAR_B;
output;

PARAM='RM'; MED_A=RM_MED1A1; MED_B=RM_MED2A; VAR_A=RM_VAR_A;
VAR_B=RM_VAR_B;
output;

PARAM='IMG'; MED_A=IMG_MED1A1; MED_B=IMG_MED2A; VAR_A=IMG_VAR_A;
VAR_B=IMG_VAR_B;
output;

PARAM='TD'; MED_A=TD_MED1A1; MED_B=TD_MED2A; VAR_A=TD_VAR_A;
VAR_B=TD_VAR_B;
output;

PARAM='L'; MED_A=L_MED1A1; MED_B=L_MED2A; VAR_A=L_VAR_A; VAR_B=L_VAR_B;
keep GR_A GR_B N1 N2 PARAM MED_A VAR_A MED_B VAR_B;
output;
run;

```

Fig. 27. Códigos SAS para criação de um arquivo de dados com para comparação dos parâmetros entre pares de grupos (Continuação).

```

/******
/*Teste t de Student para comparação de pares de tratamentos considerando variâncias
/*heterogêneas.
/******
data T_TEST;
set T_DATAS;
  EP_A=sqrt(VAR_A/N1);
  EP_B=sqrt(VAR_B/N2);
  DIF=MED_A-MED_B;
  T=DIF/(sqrt((VAR_A/N1)+(VAR_B/N2)));
  GL_NUM=((VAR_A/N1)+(VAR_B/N2))*2;
  GL_DEN=((VAR_A/N1)**2)/(N1-1)+((VAR_B/N2)**2)/(N2-1);
  GL=GL_NUM/GL_DEN;
  P_BI=2*(1-probt(abs(T),GL));
  P_DIR=1-probt(T,GL);
  P_ESQ=1-probt((-1*T),GL);
proc sort data=T_TEST;
  by PARAM;
proc print data=T_TEST noobs;
  title 'Teste t de Student para comparação entre pares de grupos';
  var PARAM GR_A GR_B MED_A EP_A MED_B EP_B P_BI P_DIR P_ESQ;
run;

```

Fig. 28. Códigos SAS para comparação entre pares de grupos de tratamentos utilizando o teste t de Student para o caso de variâncias heterogêneas.

Os testes unilaterais devem ser usados quando se conhece a priori se a provável diferença entre dois tratamentos será positiva ou negativa. Por exemplo, quando se pretende avaliar se alguma dieta reduz o potencial de crescimento populacional de uma determinada espécie, esperam-se diferenças positivas entre o grupo controle (dieta normal) e o grupo com dieta alternativa, quando se comparam as médias de Ro, Rm e Lambda. Para IMG e Dt, esperam-se diferenças negativas. Assim devem ser usados testes unilaterais à direita para o primeiro grupo de parâmetros e unilaterais à esquerda para o segundo. Em tais situações, testes unilaterais adequados têm maior poder que o correspondente teste bilateral, pois os primeiros incorporam informação sobre a suposta natureza dos efeitos dos tratamentos. Na Figura 29 são apresentados os resultados dos testes t uni e bilaterais para os dados hipotéticos onde os parâmetros associados às TBVF dos grupos 1, 2 e 3 são comparados. São apresentadas as médias de cada grupo comparado, com respectivos erros padrão e níveis de significância nominais (valores p). Quanto menores os valores p, maior a evidência contra a hipótese nula (ausência de efeito dos grupos).

PARAM	GR_A	GR_B	MED1	EP1	MED2	EP2	P_BI	P_DIR	P_ESQ
TD	1	2	4.7451	0.11834	3.6860	0.06457	0.00000	0.00000	1.00000
TD	1	3	4.7451	0.11834	3.2796	0.09128	0.00000	0.00000	1.00000
TD	2	3	3.6860	0.06457	3.2796	0.09128	0.00372	0.00186	0.99814
L	1	2	1.1572	0.00415	1.2068	0.00396	0.00000	1.00000	0.00000
L	1	3	1.1572	0.00415	1.2351	0.00718	0.00000	1.00000	0.00000
L	2	3	1.2068	0.00396	1.2351	0.00718	0.00667	0.99667	0.00333
RM	1	2	0.1460	0.00359	0.1880	0.00328	0.00000	1.00000	0.00000
RM	1	3	0.1460	0.00359	0.2112	0.00582	0.00000	1.00000	0.00000
RM	2	3	0.1880	0.00328	0.2112	0.00582	0.00630	0.99685	0.00315
RO	1	2	17.8017	1.19745	42.7111	2.53927	0.00000	1.00000	0.00000
RO	1	3	17.8017	1.19745	66.6566	7.92043	0.00075	0.99963	0.00037
RO	2	3	42.7111	2.53927	66.6566	7.92043	0.02284	0.98858	0.01142
IMG	1	2	19.7379	0.10081	19.9798	0.10934	0.12262	0.93869	0.06131
IMG	1	3	19.7379	0.10081	19.9188	0.22230	0.47858	0.76071	0.23929
IMG	2	3	19.9798	0.10934	19.9188	0.22230	0.81123	0.40562	0.59438

Fig. 29. Resultados dos testes t uni e bilaterais para comparação entre todos os pares (GR A, GR B) formados pelos grupos 1, 2 e 3, com relação aos parâmetros associados às TBVF.

Anexos

Observação: Os arquivos dos anexos estão todos disponíveis para download no site da Embrapa Meio Ambiente no seguinte endereço: <http://www.cnpma.embrapa.br/public/comunicado33.html> no link "Anexos".

Anexo 1. Programa SAS para comparação de parâmetros associados às tabelas de vida e fertilidade (versão em português): TabelaDeVida.sas.

Anexo 2. Arquivo Exel® (Exemplo.xls) contendo os dados para estimação de parâmetros associados às tabelas de vida e fertilidade de artrópodes.

Anexo 3. Número de ovos por fêmea (OVTOTAL) em cada grupo, durante todo o período de oviposição (OviposicaoExemplo.xls).

Anexo 4. Tabelas de vida e fertilidade para os grupos 1, 2 e 3, geradas a partir dos dados do arquivo "Exemplo.xls" (TabelaDeVidaExemplo.xls).

Anexo 5. Pseudovalores das estimativas dos parâmetros associados às tabelas de vida e fertilidade para os grupos 1, 2 e 3 (PseudovaloresExemplo.xls).

Referências Bibliográficas

BLEICHER, E.; PARRA, J. R. P. Espécies de *Trichogramma* parasitóides de *Alabama argillacea*. II. Tabela de vida e fertilidade e parasitismo de três populações. **Pesquisa Agropecuária Brasileira**, v.25, p.207-214, 1990.

BRODSGAARD, H. F. Effect of photoperiod on the bionomics of *Frankliniella occidentalis* (Pergand) (Thysanoptera, Thripidae). **Journal of Applied Entomology**, v.17, p.498-507, 1994.

CHAKRABORTY, N.; BHATTACHARYA, S.S.; SAHAKUNDU, A.K.; SEN, S.K. Lifetable of uzifly *Exorista bombycis* Louis (Diptera Tachinidae) parasitizing 2nd stage larvae of silkworm, *Bombix mori*. **Entomon**, v.21, p.243-250, 1996.

COATS, S. A. Life cycle and behavior of *Muscidifurax zaraptor* (Hymenoptera: Pteromalidae). **Annals of the Entomological Society of America**, v.69, p.772-779, 1976.

EFRON, B. **The Jackknife, the Bootstrap and other resampling plans**. Philadelphia: Society for Industrial and Applied Mathematics, 1983.

FERNANDEZ-CASALDERREY, A.; FERRANDO, M.D.; ANDREAU-MOLINER, E. Effect of sublethal diazinon concentrations on the demographic parameters of *Brachionus calyciflorus* Pallas (Rotifera). **Bulletin of Environmental Contamination and Toxicology**, v.48, p.202-208, 1992.

HANCE, T.; NIBELLE, D.; LEBRUN, P.; VAN IMPE, G.; VAN HOVE, C. Selection of *Azolla* forms resistant to the water lily aphid, *Rhopalosiphum nymphaeae*. Life history of *Rhopalosiphum nymphaeae*. **Entomologia Experimentalis et Applicata**, v.70, p.11-17, 1994.

HINDE, J.; DEMÉTRIO, C. Overdispersion: models and estimation. **Computational Statistics and Data Analysis**, v.27, p.151-170, 1998.

HULTING, F. L.; ORR, D.B.; OBRYCKI, J.J. A computer program for calculation and statistical comparison of intrinsic rates of increase and associated life table parameters. **Florida Entomologist**, v.73, p.600-612, 1990.

MAIA, A. de H. N.; LUIZ, A. J. B., CAMPANHOLA, C. Statistical inference on associated fertility life table parameters using jackknife technique: computational aspects. **Journal of Economic Entomology**, v.93, n.2, p.511-518, 2000.

MANLY, B.F.J. **Randomization tests and Monte Carlo methods in biology**. London: Chapman & Hall, 1991.

MEYER, J.S.; ITERSOLL, C.G.; MACDONALD, L.L.; BOYCE, M.S. Estimating uncertainty in population growth rates: Jackknife vs. Bootstrap techniques. **Ecology**, v.67, p.1156-1166, 1986.

MONTGOMERY, D. C. **Design and analysis of experiments**. New York: John Wiley, 1991.

NARDO, E.A.B.; MAIA, A.H.N.; WATANABE, M.A. Effect of a formulation of *Anticarsia gemmatilis* (Lepidoptera: Noctuidae) on the predator *Podisus nigrispinus* (Heteroptera: Pentatomidae: Asopinae) using fertility life table parameters. **Environmental Entomology**, v.30, n.6, p.1164-1173, 2001.

NASCIMENTO, M.L.; CAPALBO, D.F.; MORAES, G.J.; NARDO, E.A.B.; MAIA, A.H.N.; OLIVEIRA, R.C.A.L. Effect of a formulation of *Bacillus thuringiensis* Berliner var. *kurstaki* on *Podisus nigrispinus* Dallas (Heteroptera: Pentatomidae asopinae). **Journal of Invertebrate Pathology**, v.72, p.178-180, 1998.

PARRA, J.R.P.; HADDAD, M. de L.; SILVEIRA-NETO, S. Tabela de vida de fertilidade de *Perileucoptera coffeella* (Guerin- Meneville, 1842) (Lepidoptera, Lyonetiidae) em três temperaturas. **Revista Brasileira de Entomologia**, v.39, p.125-129, 1995.

SAS INSTITUTE INC. **SAS language**: Reference, version 6. 4. ed. Cary: SAS Institute, 1990. v.2.

SHARMA, K.C.; CHAUHAN, U.; VERMA, A.K. Life table for the black bean bug: *Chauliops nigrescens* Distant (Hemiptera : Lygaeidae) on French bean. **Annals of Agricultural Research**, v.15, p.191-193, 1994.

SHARMA, K.C.; CHAUHAN, U.; VERMA, A.K. Life-fertility table of the pea leafminer *Chromatomyia horticola* (Goureau) (Diptera : Agromyzidae) on pea, *Pisum sativum*, L. **Annals of Agricultural Research**, v.18, p.25-28, 1997.

SIDÁK, Z. Rectangular confidence regions for the means of multivariate normal distributions. **Journal of the American Statistical Association**, v.62, p.626-633, 1967.

SOUTHWOOD, T.R.E. **Ecological methods with particular reference to the study of insect population**. 2. ed. London: Chapman & Hall, 1978.

TUKEY, J.W. Bias and confidence in not quite large samples. **Annals of Mathematics and Statistics**, v.29, p.614, 1958.

ZAR, J. H. **Biostatistical analysis**. 2. ed. New Jersey: Prentice-Hall, 1984.

Comunicado Técnico, 33

Ministério da
Agricultura, Pecuária
e Abastecimento



Exemplares desta edição podem ser adquiridos na:
Embrapa Meio Ambiente
Endereço: Rodovia SP 340 km 127,5
Caixa Postal 69, Tanquinho Velho
13.820-000 Jaguariúna/SP
Fone: (19) 3867-8700
Fax: (19) 3867-8740
E-mail: sac@cnpmma.embrapa.br

1ª edição eletrônica
2006

Comitê de Publicações

Presidente: Ladislau Araújo Skorupa.
Secretário-Executivo: Sandro Freitas Nunes.
Bibliotecário: Maria Amélia de Toledo Leme.
Membros: Cláudio César de A. Buschinelli, Heloisa Ferreira Filizola, Manoel Dornelas de Souza, Maria Conceição P. Young Pessoa, Marta Camargo de Assis, Osvaldo Cabral

Expediente

Tratamento das ilustrações: Sandro Freitas Nunes.
Editoração eletrônica: Sandro Freitas Nunes.